
Computational Methods 
 

DSP Programming – Demonstration Program 4 

 

' FBankConsole - demonstrate application of filterbank to wav file 

' 

' Select a Console application. 

' 

' Use the "My Project" dialog to configure the program: 

' - under the Compile tab, press "Advanced Compilation Options" 

'   and change the Target CPU to "x86" 

' - under the References tab, click the "Add" button then browse for 

'   and select the BasicDSP.dll library. 

' 

' Run from Command Prompt or CYGWIN: 

'   FBankConsole input.wav >output.txt 

' 

Imports BasicDSP 

Imports BasicDSP.Filter 

' 

Module FBankConsole 

 

    ' number of channels 

    Private Const NCHAN As Integer = 19 

    ' channel edge frequencies (Hz) 

    Dim cutfreq() As Double = {180, 300, 420, 540, 660, 780, 900, 1075, 

1225, 1375, 1525, 1700, 1900, 2100, 2300, 2550, 2850, 3150, 3475, 4000} 

    ' output rate of filterbank (frames/sec) 

    Private Const ORATE As Double = 100 

 

    Sub Main(ByVal args() As String) 

        Dim chan As Integer 

 

        ' check arguments 

        If args.Length = 0 Then 

            Console.WriteLine("usage: fbankconsole input.wav") 

            Environment.Exit(1) 

        End If 

 

        ' read in a WAV audio file 

        Dim wv As New Signal 

        If Not wv.LoadWaveFile(args(0)) Then 

            Console.WriteLine("could not open file: " & args(0)) 

            Environment.Exit(1) 

        End If 

 

        ' build the bank of analysis filters 

        Dim fbank(NCHAN) As LTISystemChain 

        For chan = 1 To NCHAN 

            fbank(chan) = ButterworthBandPass(cutfreq(chan - 1) / wv.Rate, 

cutfreq(chan) / wv.Rate, 4) 

        Next 

 

        ' build a bank of smoothing filters at half output rate 

        Dim sbank(NCHAN) As LTISystemChain 

        For chan = 1 To NCHAN 

            sbank(chan) = ButterworthLowPass((0.5 * ORATE) / wv.Rate, 4) 

        Next 

        ' process the signal sample by sample 

        Dim t As Double = 0 

        Dim out(NCHAN) As Double 

        For n As Integer = wv.First To wv.Last 

            ' for each channel 

            For chan = 1 To NCHAN 

                ' filter the input sample 

                out(chan) = fbank(chan)(wv(n)) 

                ' rectify 

                out(chan) = Math.Abs(out(chan)) 

                ' smooth 

                out(chan) = sbank(chan)(out(chan)) 

            Next 

            t += wv.Period 

            ' every so often, output frame 

            If (t > 1 / ORATE) Then 

                For chan = 1 To NCHAN 

                    Console.Write("{0,6:F02} ", 20 * Math.Log10(out(chan))) 

                Next 

                Console.WriteLine() 

                t -= 1 / ORATE 

            End If 

        Next 

    End Sub 

 

End Module 

 

 

 

 


